
Survey Data Analysis

Introduction to Stata

About StataCorp

Founded in 1982 in Santa Monica, CA, under the name CRC
Bill Gould and Finis Welch, UCLA
Sold time on a mainframe
Stata 1.0 released January 1985
Gave up mainframe business in 1986

Relocated to College Station, TX in 1993

Changed name to Stata Corporation at that time
Later created Stata Press, a division of StataCorp

Survey Data Analysis

Introduction to Stata

Stata software

Pronounce it anyway you like. We say it should rhyme with
“data”.
Stata is a name, not an acronym (we do not use STATA)

Available on many platforms
Mac
Windows
Unix

Linux
IBM AIX
Oracle Solaris

Survey Data Analysis

Introduction to Stata

Stata software

Stata 12 has been announced and will be shipping soon
Major versions (e.g., Stata 9, Stata 10, Stata 11) are sold
Minor versions (e.g., Stata 10.1, Stata 11.2) are free
updates
Other additions/fixes are also free updates
Updates are done over the web

Survey Data Analysis

Introduction to Stata

Documentation

Over 9,500 pages of documentation in Stata 12

Organized in fifteen volumes

[GS] Getting Started–Mac, Unix, Windows
[U] User’s Guide
[D] Data Management
[G] Graphics

[MI] Multiple Imputation
[MV] Multivariate Statistics

[R] Base Reference (4 volumes)
[ST] Survival Analysis/Epidemiological Tables

[SEM] Structural equations modeling
[SVY] Survey Data

[TS] Time Series
[XT] Panel/Longitudinal Data

[P] Programming

Introduction to Stata

This is Stata 12

J. Pitblado (StataCorp) Bamberg, Germany 2011 7 / 104



Survey Data Analysis

Introduction to Stata

Graphical User Interface (GUI)

You can point and click all the way
Main menus are Data, Graphics, and Statistics
Filling out a dialog box generates the needed command
As such, it is a great way to learn Stata
You can choose to work interactively by typing commands
and/or using the menus
You can also work through editable scripts of commands,
known as do-files

Survey Data Analysis

Introduction to Stata

Graphical User Interface (GUI)

Survey Data Analysis

Introduction to Stata

Graphical User Interface (GUI)

Survey Data Analysis

Introduction to Stata

Publication-quality graphics

Hawaii

% homes cost $100K+

Source: A Visual Guide to Stata Graphics

Combined 1980 and 1990
U.S. Census Data

J. Pitblado (StataCorp) Bamberg, Germany 2011 11 / 104



Survey Data Analysis

Introduction to Stata

Why use Stata for survey data?

Stata is fully “survey-capable”
In Stata, there is a clear separation between setting the
design and performing the actual analysis
You declare the design characteristics using svyset

This declaration is a one-time event. You save the survey
settings along with the data
You perform the analysis just as you would with i.i.d. data –
you just have to add the svy: prefix

Survey Data Analysis

Introduction to Stata

Let’s get started

First of all, let’s make sure our Stata is up-to-date:

. update query

and follow the instructions. You want to get in the habit of
doing this.
Now let’s begin a log of what we are doing

. log using lesson1

At the end of this lesson, we will close the log by typing

. log close

Survey Data Analysis

Introduction to Stata

Stata datasets

Stata datasets carry the .dta extension, and these are
binary files
The format is the same across all platforms
To work with a dataset, it must be loaded into active
memory from disk

. use "C:\program files\stata11\auto"

or, better yet, from the web

. use http://www.stata-press.com/data/r11/cancer

The above is so useful, we even have a shorthand for it,
which we will use quite often

. webuse cancer

Survey Data Analysis

Introduction to Stata

Getting your data into Stata

You can use infile and insheet to import ASCII files of
any complexity
fdause and fdasave will load and save SAS transport
files (.xpt)
xmluse and xmlsave for .xml files
Stat/Transfer by Circle Systems, Inc., is a very handy
commercial program
By far the easiest is to Copy and Paste your Excel
spreadsheets into the Data Editor

Survey Data Analysis

Introduction to Stata

Getting the help you need

Help for any Stata command, e.g.

. help logit

Manuals are available in PDF files, and can be accessed
from the online help
Help from the top level

. help contents

Help from Stata and the web

. findit survey data

. findit violin plots

findit is like Google for Stata. You can even install new
software from it

Survey Data Analysis

Introduction to Stata

Getting the help you need

We are now finished with this lesson.

. log close

. view lesson1.smcl

Survey Data Analysis

Fitting models

Start a new log for this lesson.

. log using lesson2

Survey Data Analysis

Fitting models

Overview

We call commands that fit models estimation commands
Uniformity across all models is critical:

syntax
displayed results
returned results
predictions and diagnostics
testing and inference

Learn to use one, learn them all
Survey-design aspects layer on top of the principles behind
estimation

Survey Data Analysis

Fitting models

Syntax

Estimation commands follow a standard syntax

command varlist if expression in range , options

varlist specifies the model, usually dependent variable
followed by a set of regressors
if and in determine the sample used to fit the model
options are general options controlling the estimation and
displaying of results, or, model-specific options such as
how to handle ties in a Cox regression

Survey Data Analysis

Fitting models

Logistic Regression

Example
The Tower of London (Rabe-Hesketh et al. 2001)

Study of cognitive abilities of patients with schizophrenia
Cognitive ability was measured by successful completion
of the Tower of London, a computerized task (binary
variable dtlm)
226 subjects, all but one tested at three difficulty levels
(variable difficulty)
Subjects were not only patients (group==3), but relatives
(group==2) and nonrelated controls (group==1)
We can thus propose a logistic regression model for dtlm
as, initially, a function of difficulty

Survey Data Analysis

Fitting models

Using logit

Let’s have a look at the dataset:

. webuse towerlondon, clear

. describe

. notes

. codebook

Let’s now fit the logistic model:

. logit dtlm difficulty

How about odds ratios and 90% CIs?

. logit, or level(90)

Survey Data Analysis

Fitting models

Categorical variables

Variable group has three levels, and the whole point of the
study is to assess the affect of this factor

. logit dtlm difficulty i.group

or the interaction effects of group and difficulty

. gen diff2 = difficulty + 1

. logit dtlm group#diff2, or

. logit dtlm group##diff2, or

. logit dtlm group#c.difficulty, or

For models with a single factor with a small number of
levels, you might prefer dummy variables to abstract
codings

. describe

. gen relative = group == 2

. gen sphrenic = group == 3

. logit dtlm difficulty relative sphrenic, or

Survey Data Analysis

Fitting models

Weights

Stata supports four kinds of weights

fweight – Frequency
aweight – Analytic
pweight – Probability/Sampling
iweight – Importance

Use pweight for survey data
In our example, suppose we wanted to weight according to
family size:

. bysort family: gen fsize = _N

. logit dtlm difficulty relative sphrenic [pw=fsize], or

Note that Stata uses Taylor linearization to estimate
standard errors when pweights are specified

Survey Data Analysis

Fitting models

Clustering

In this example, each subject takes the Tower of London
test three times, at three levels of difficulty
There is likely intra-subject correlation to account for. You
can either model it directly, or control for it the “survey” way

. logit dtlm difficulty relative sphrenic [pw=fsize], or ///

> vce(cluster subject)

It would probably be more appropriate to cluster on
family, since the patients and their relatives are likely to
be correlated

. logit dtlm difficulty relative sphrenic [pw=fsize], or ///

> vce(cluster family)

Clustering on family subsumes clustering on subject.

Survey Data Analysis

Fitting models

Cataloging model fits

Within a Stata session, you can store estimates for later
use and for comparison with other models

. logit dtlm difficulty relative sphrenic

. est store logit

. probit dtlm difficulty relative sphrenic

. est store probit

. est dir

. est table _all

. est stats _all

Stata 10 introduced the ability to also save results to disk
for retrieval during a later session

. probit

. est save probit

The next time you start Stata, you could type

. est use probit

. probit

Survey Data Analysis

Fitting models

Predictions and Diagnostics

The predict command will generate a new variable
containing the prediction or diagnostic of your choice (or
the default)

. logit dtlm difficulty relative sphrenic

. predict phat

. list dtlm difficulty relative sphrenic phat in 1/20

For logistic regression a useful diagnostic plot is one of
Pearson residuals vs. the linear predictor (the log relative
odds)

. predict pearson, residuals

. predict xbeta, xb

. scatter pearson xbeta

Predictions can be made on the estimation data, or on
other data as long as the variable names are the same

Survey Data Analysis

Fitting models

Tests of hypotheses

As part of any estimation results, you get an omnibus test
for all model coefficients. This omnibus test is either a
likelihood-ratio test, or a Wald test

. logit

You can use test to get other Wald tests

. test sphrenic = -1

. test relative sphrenic

You can use lrtest to get other LR tests, but this requires
refitting the model

. est store full

. logit dtlm difficulty

. lrtest . full

LR tests not appropriate for survey data; don’t worry, Stata
will tell you so

Fitting models

Tests of hypotheses

We are now finished with this lesson.

. log close

Survey Data Analysis

Types of data

Why survey data?

Collecting data can be expensive and time consuming.
Consider how you would collect the following data:

Smoking habits of teenagers
Birth weights for expectant mothers with high blood
pressure

Using stages of clustered sampling can help cut down on
the expense and time.

Survey Data Analysis

Types of data

Types of data

Simple random sample (SRS) data
Observations are ”independently” sampled from a data
generating process.

Typical assumption: independent and identically distributed
(iid)
Make inferences about the data generating process
Sample variability is explained by the statistical model
attributed to the data generating process

Standard data
We’ll use this term to distinguish this data from survey data.

Survey Data Analysis

Types of data

Types of data

Correlated data
Individuals are assumed not independent.

Observations are taken over time
Random effects assumptions
Cluster sampling

What do you do about it?

Time-series models
Longitudinal/panel data models
vce(cluster ...) option

Survey Data Analysis

Types of data

Types of data

Survey data
Individuals are sampled from a fixed population according to a
survey design.

Distinguishes itself from other forms of data:

Complex nature under which individuals are sampled
Make inferences about the fixed population
Sample variability is attributed to the survey design

Aspects of survey data

Start a new log for this lesson.

. log using lesson4

Survey Data Analysis

Aspects of survey data

Survey data characteristics

Standard data
Estimation commands for standard data:

proportion
regress

We’ll refer to these as standard estimation commands.

Survey data
Survey estimation commands are governed by the svy
prefix.

svy: proportion
svy: regress

svy requires that the data is svyset.

Survey Data Analysis

Aspects of survey data

Survey in Stata is easy

Once you get the design aspects and other preferences
declared, estimation is quite easy.
For example, to estimate proportions:

. webuse nhanes2, clear

. proportion sex

. svyset

. svy: proportion sex

So really, this workshop is about declaring your design to
Stata, and for that we have svyset

Aspects of survey data

Some definitions

Single-stage syntax

svyset
[
psu

] [
weight

] [
,strata(varname)fpc(varname)

]
PSU – primary sampling units
pweight – sampling weights
Strata
FPC – finite population correction

Survey Data Analysis

Aspects of survey data

Some definitions

Sampling unit
An individual or collection of individuals from the population that
can be selected for observation.

Sampling groups of individuals is synonymous with cluster
sampling.
Cluster sampling usually results in inflated variance
estimates compared to SRS.

Survey Data Analysis

Aspects of survey data

Some definitions

Sampling weight
The reciprocal of the probability for an individual to be sampled.

Probabilities are derived from the survey design.

Sampling units
Strata

Typically considered to be the number of individuals in the
population that a sampled individual represents.
Reduces bias induced by the sampling design.

Survey Data Analysis

Aspects of survey data

Some definitions

Strata
In stratified designs, the population is partitioned into
well-defined groups, called strata.

Sampling units are independently sampled from within
each stratum.
Stratification usually results in smaller variance estimates
compared to SRS.
Although there is potential for improving efficiency by
reducing sampling variability, it is usually not very practical
to stratify on demographic information.

Survey Data Analysis

Aspects of survey data

Some definitions

Finite population correction (FPC)
An adjustment applied to the variance due to sampling without
replacement.

Sampling without replacement from a finite population
reduces sampling variability.

Aspects of survey data

What does sampling look like?

Population 1000

Aspects of survey data

What does sampling look like?

SRS sample 200

Aspects of survey data

What does sampling look like?

Cluster sample 20 (208 obs)

Survey Data Analysis

Aspects of survey data

What does sampling look like?

Stratified sample 198

Aspects of survey data

What does sampling look like?

Stratified-cluster sample 20 (215 obs)

Aspects of survey data

What does sampling look like?

We are now finished with this lesson.

. log close

Using svyset

Start a new log for this lesson.

. log using lesson5

Survey Data Analysis

Using svyset

Recall the single-stage syntax

Single-stage syntax

svyset
[
psu

] [
weight

] [
,strata(varname)fpc(varname)

]
PSU – primary sampling units
pweight – sampling weights
Strata
FPC – finite population correction

Survey Data Analysis

Using svyset

Some examples

Simple Random Sample:

. sysuse auto, clear

. svyset _n

. svy: regress mpg weight

Stratified design (National Maternal and Infant Health
Survey):

. webuse nmihs, clear

. describe

. svyset [pw=finwgt], strata(stratan)

. svy: tabulate agegrp lowbw

Single-stage design with all the characteristics:

. webuse fpc, clear

. describe

. svyset psuid [pw=weight], strata(stratid) fpc(Nh)

Survey Data Analysis

Using svyset

A multistage design

High school data

Purpose: Study the smoking habits of teenagers in the US.

Multistage design:

1 Use state for strata, and counties are the PSUs.
2 The second stage units are high schools, randomly

selected within each sampled county.
3 Stratifying on gender, the final stage units are high school

seniors, randomly selected within each sampled high
school.

Using svyset

A multistage design

Multistage syntax

svyset psu
[
weight

] [
, strata(varname) fpc(varname)

][
|| ssu

[
, strata(varname) fpc(varname)

] ][
|| ssu

[
, strata(varname) fpc(varname)

] ]
...

Stages are delimited by “||”
SSU – secondary/subsequent sampling units
FPC is required at stage s for stage s + 1 to play a role in
the linearized variance estimator

Survey Data Analysis

Using svyset

A multistage design

Multiple stages of cluster sampling

1 PSUs are independently selected within each Stratum.
2 SSUs are independently selected within each sampled

PSU.
3 . . .

Sampling units are independently selected within each
sampled SSU.
Stratification is also allowed at each sampling stage.

Using svyset

A multistage design

High school data

Smoking habits of teenagers in the US.

1 Counties are randomly selected within each State.
2 High schools are randomly selected within each sampled

county.
3 Female and male seniors are randomly selected within

each sampled high school.

Using svyset

A multistage design

FPC variables
ncounties – number of counties within each category of
state

nschools – high schools within state county

nseniors – high school seniors within state county
school sex

. webuse seniors

. svyset county [pw=sampwgt], strata(state) fpc(ncounties) ///

> || school, fpc(nschools) ///

> || _n, strata(gender) fpc(nseniors)

. save myseniors

. svy: logit smoked i.gender i.race, or

. test 2.race 3.race

J. Pitblado (StataCorp) Bamberg, Germany 2011 55 / 104

Using svyset

Poststratification

Poststratification
A method for adjusting sampling weights, usually to account for
underrepresented groups in the population.

Adjusts weights to sum to the poststratum sizes in the
population
Reduces bias due to nonresponse and underrepresented
groups
Can result in smaller variance estimates

Syntax

svyset ... poststrata(varname) postweight(varname)

Using svyset

Poststratification

Cats and dogs

Source: Levy and Lemeshow (1999)
Veterinarian has 1300 clients, 450 cats and 850 dogs
He wishes to estimate average annual expenses, but only
has time to randomly select 50 clients (i.e., each client
represents 26)
Problem: As we shall see, dogs are about twice as
expensive as cats

Using svyset

Poststratification

Let’s estimate total expenses:

. webuse poststrata

. describe

. bysort type: sum totexp

. codebook weight

. svyset [pw=weight]

. svy: mean totexp

. codebook postwgt fpc

. svyset [pw=weight], poststrata(type) postweight(postwgt) fpc(fpc)

. svy: mean totexp

. svy: total totexp

Using svyset

Strata with a single sampling unit

Problem
This is a big issue for variance estimation:

Consider a sample with only 1 observation
svy reports missing standard error estimates by default

Solution
Use svydescribe:

Describes the strata and sampling units
Helps find strata with a single sampling unit

Drop them from the estimation sample.
svyset one of the ad-hoc adjustments in the
singleunit() option.
Somehow combine them with other strata.

Survey Data Analysis

Using svyset

Strata with a single sampling unit

Example: Second National Health and Nutrition
Examination Survey

. webuse nhanes2, clear

. svyset

. svydescribe

Everything looks fine, but what if we are examining high
density lipids?

. codebook hdresult

. svy: mean hdresult

. svydescribe if e(sample), single

Even better would be

. svydescribe hdresult, single

Using svyset

Certainty units

Certainty units
Sampling units that are guaranteed to be chosen by the design.

Treat each certainty unit as a stratum with an FPC of 1.
No contribution to the variance.
Certainty PSUs are not counted in the design degrees of
freedom.

. svyset

. svyset psu [pw=finalwgt], strata(strata) singleunit(certainty)

. svy: mean hdresult

Using svyset

Certainty units

We are now finished with this lesson.

. log close

Variance estimation

Start a new log for this lesson.

. log using lesson6

Variance estimation

Stata has five variance estimation methods for survey data:

Linearization
Balanced repeated replication (BRR)
Survey jackknife
Survey bootstrap
Successive difference replication (SDR)

Survey Data Analysis

Variance estimation

Linearization

Linearization
A method for deriving a variance estimator using a first order
Taylor approximation of the point estimator of interest.

Foundation: Variance of the total estimator

Syntax

svyset ...
[
vce(linearized)

]
Delta method
Huber/White/robust/sandwich estimator

Variance estimation

The total estimator

Total estimator – Stratified two-stage design
yhijk – observed value from a sampled individual
Strata: h = 1, . . . ,L
PSU: i = 1, . . . ,nh

SSU: j = 1, . . . ,mhi

Individual: k = 1, . . . ,mhij

Ŷ =
∑

whijkyhijk

V̂ (Ŷ ) =
∑

h

(1− fh)
nh

nh − 1

∑
i

(yhi − yh)
2 +

∑
h

fh
∑

i

(1− fhi)
mhi

mhi − 1

∑
j

(yhij − yhi)
2

Variance estimation

The total estimator

Returning to our seniors data, we can estimate the total
number of seniors who have smoked

. use myseniors, clear

. svyset

. svy: total smoked

Prior to Stata 9, you could only incorporate the first stage
of the sample design

. svyset county [pw=sampwgt], strata(state) fpc(ncounties)

. svy: total smoked

Survey Data Analysis

Variance estimation

Regression models

Linearized variance for regression models
Model is fit using estimating equations.

Ĝ() is a total estimator, use Taylor expansion to get V̂ (β̂).

Ĝ(β) =
∑

j

wjsjxj = 0

V̂ (β̂) = DV̂{Ĝ(β)}|
β=bβD′

Survey Data Analysis

Variance estimation

Example

For this, we return to NHANES2

. webuse nhanes2, clear

. svyset

and fit a logit model for high blood pressure:

. describe highbp height weight age female race

. discard

. local model highbp height weight c.age##c.age i.female i.race

. svy: logit `model´, baselevel

. est store taylor

We can also estimate the odds ratio for a 5-year age
increase and a 10 kg weight increase, and its
survey-adjusted standard error

. lincom 5*age + 25*c.age#c.age + 10*weight, or

Variance estimation

Balanced repeated replication (BRR)

Balanced repeated replication
For designs with two PSUs in each of L strata.

Compute replicates by dropping a PSU from each stratum.
Find a balanced subset of the 2L replicates. L ≤ r < L + 4
The replicates are used to estimate the variance.

Syntax

svyset ... vce(brr)
[
mse

]

Variance estimation

Balanced repeated replication (BRR)

BRR replicate weights

wj – sampling weight for individual j , in the first PSU of
stratum h.
Hr is a Hadamard matrix for r replications; H ′r Hr = r I.
Fay’s adjustment f ; f = 0 by default.

The adjusted sampling weight for the i th replicate is

w∗j =

{
fwj , if Hr [i ,h] = −1

(2− f )wj , if Hr [i ,h] = +1

Variance estimation

Balanced repeated replication (BRR)

BRR variance formulas

θ̂ – point estimates
θ̂(i) – i th replicate of the point estimates

θ(.) – average of the replicates

Default variance formula:

V̂ (θ̂) =
1

r(1− f )2

r∑
i=1

{θ̂(i) − θ(.)}{θ̂(i) − θ(.)}′

Mean squared error (MSE) formula:

V̂ (θ̂) =
1

r(1− f )2

r∑
i=1

{θ̂(i) − θ̂}{θ̂(i) − θ̂}′

Survey Data Analysis

Variance estimation

Balanced repeated replication (BRR)

We can use a version of NHANES2 that already has a set of
replicate-weight variables in it

. webuse nhanes2brr, clear

. svyset, vce(brr, mse) noclear

. svy: logit `model´, baselevel

We can also compare with the previous results that used
Taylor linearization

. est store brr

. est table _all, se eform

Survey Data Analysis

Variance estimation

Jackknife

The jackknife
A replication method for variance estimation. Not restricted to a
specific survey design.

Delete-1 jackknife: drop 1 PSU
Delete-k jackknife: drop k PSUs within a stratum

Syntax

svyset ... vce(jackknife)
[
mse

]

Survey Data Analysis

Variance estimation

Jackknife

Delete-1 jackknife replicate weights

whij – sampling weight for individual j in PSU i of stratum h.
Dropping PSU i∗ from stratum h∗.
nh∗ replicates from stratum h∗.

The adjusted sampling weight is

w∗hij =


0 , if h = h∗ and i = i∗

nh
nh−1whij , if h = h∗ and i 6= i∗

whij , otherwise

Survey Data Analysis

Variance estimation

Jackknife

Delete-k jackknife replicate weights

whij – sampling weight for individual j in PSU i of stratum h.
Drop k PSUs from stratum h∗.
ch∗ =

(nh∗
k

)
replicates from stratum h∗.

The adjusted sampling weight is

w∗hij =


0 , if h = h∗ and i is dropped

nh
nh−k whij , if h = h∗ and i is not dropped

whij , otherwise

Survey Data Analysis

Variance estimation

Jackknife

Jackknife variance formulas

θ̂(h,i) – replicate of the point estimates from stratum h, PSU
i
θh – average of the replicates from stratum h
mh = (nh − 1)/nh – delete-1 multiplier for stratum h
mh = (nh − k)/chk – delete-k

Default variance formula:

V̂ (θ̂) =
L∑

h=1

(1− fh) mh

nh∑
i=1

{θ̂(h,i) − θh}{θ̂(h,i) − θh}′

Mean squared error (MSE) formula:

V̂ (θ̂) =
L∑

h=1

(1− fh) mh

nh∑
i=1

{θ̂(h,i) − θ̂}{θ̂(h,i) − θ̂}′

Survey Data Analysis

Variance estimation

Example

You can specify jackknife standard errors when you fit the
model:

. webuse nhanes2, clear

. svyset

. svy jackknife, mse: logit `model´, baselevel

or you can, equivalently, specify jackknife estimation as
your preferred method

. svyset psu [pw=finalwgt], strata(strata) vce(jackknife, mse)

. svy: logit `model´, baselevel

Survey Data Analysis

Variance estimation

Bootstrap

The bootstrap
Even less restrictive on the design and parameters than the
delete-1 jackknife.

Resample the observed data by adjusting the sampling
weights.
Requires replicate weight variables.

Syntax
svyset ... vce(bootstrap) bsrweight(varlist)[

bsn(#) mse
]

Survey Data Analysis

Variance estimation

Bootstrap

Bootstrap variance formulas

θ̂ – point estimates
θ̂(i) – i th replicate of the point estimates

θ(.) – average of the replicates
b – number of bootstrap samples used to generate each
replicate weight variable, default is bsn(1)

Default variance formula:

V̂ (θ̂) =
b
r

r∑
i=1

{θ̂(i) − θ(.)}{θ̂(i) − θ(.)}′

Mean squared error (MSE) formula:

V̂ (θ̂) =
b
r

r∑
i=1

{θ̂(i) − θ̂}{θ̂(i) − θ̂}′

Survey Data Analysis

Variance estimation

Successive difference replication

Successive difference replication – SDR
Replication method designed for systematic samples where the
observed sampling units are ordered.

Resample the observed data by adjusting the sampling
weights.
Requires replicate weight variables.

Syntax

svyset ... vce(sdr) sdrweight(varlist)
[
mse

]

Survey Data Analysis

Variance estimation

Successive difference replication

SDR variance formulas

θ̂ – point estimates
θ̂(i) – i th replicate of the point estimates

θ(.) – average of the replicates
f – sampling fraction from fpc() option

Default variance formula:

V̂ (θ̂) = (1− f )
4
r

r∑
i=1

{θ̂(i) − θ(.)}{θ̂(i) − θ(.)}′

Mean squared error (MSE) formula:

V̂ (θ̂) = (1− f )
4
r

r∑
i=1

{θ̂(i) − θ̂}{θ̂(i) − θ̂}′

Survey Data Analysis

Variance estimation

Replicate weights

Replicate weight variable
A variable in the dataset that contains sampling weight values
that were adjusted for resampling the data.

Typically used to protect the privacy of the survey
participants.
Eliminate the need to svyset the strata and PSU
variables.

Syntax

svyset ... brrweight(varlist)
[
fay(#)

]
svyset ... jkrweight(varlist

[
, ... multiplier(#)

]
)

svyset ... bsrweight(varlist)
[
bsn(#)

]
svyset ... sdrweight(varlist)

Survey Data Analysis

Variance estimation

Example

Consider a privacy-conscious version of NHANES:

. webuse nhanes2jknife, clear

. svyset

. webuse nhanes2jknife, clear

. describe

. svyset [pw=finalwgt], vce(jackknife) jkrweight(jkw *)

We can thus “replicate” the jackknife calculation, without
having to know anything about strata and PSU
membership

. svy, mse: logit `model´, baselevel

Survey Data Analysis

Variance estimation

Example

We are now finished with this lesson.

. log close

Survey Data Analysis

Subpopulations

Start a new log for this lesson.

. log using lesson7

J. Pitblado (StataCorp) Bamberg, Germany 2011 86 / 104



Survey Data Analysis

Subpopulations

Two perspectives sampling subsets

Focus on a subset of the population
Subpopulation variance estimation:

Assumes the same survey design for subsequent data
collection.
The subpop() option.

Restricted-sample variance estimation:

Assumes the identified subset for subsequent data
collection.
Ignores the fact that the sample size is a random quantity.
The if and in restrictions.

Survey Data Analysis

Subpopulations

The subpopulation total

Total from SRS data
Data is y1, . . . , yn and S is the subset of observations.

δj(S) =

{
1, if j ∈ S
0,otherwise

Subpopulation (or restricted-sample) total:

ŶS =
n∑

j=1

δj(S)wjyj

Sampling weight and subpopulation size:

wj =
N
n
, NS =

n∑
j=1

δj(S)wj =
N
n

nS

Survey Data Analysis

Subpopulations

The subpopulation total

Variance of a subpopulation total
Sample n without replacement from a population comprised of
the NS subpopulation values with N − NS additional zeroes.

V̂ (ŶS) =
(

1− n
N

) n
n − 1

n∑
j=1

{
δj(S)wjyj −

1
n

ŶS

}2

Variance of a restricted-sample total
Sample nS without replacement from the subpopulation of NS
values.

Ṽ (ŶS) =

(
1− nS

N̂S

)
nS

nS − 1

n∑
j=1

δj(S)

{
yj −

1
nS

ŶS

}2

Survey Data Analysis

Subpopulations

Example

Returning the the National Maternal and Infant Health
Survey

. webuse nmihs, clear

. svyset

. des birthwgt highbp

. label list hibp

We can estimate the mean birth weight for the
high-blood-pressure subpopulation

. svy, subpop(highbp): mean birthwgt

How does that compare to the restricted-sample estimate?

. svy: mean birthwgt if highbp

Survey Data Analysis

Subpopulations

Getting fancy

How about the other subpopulation, those without high
blood pressure?

. svy, subpop(if !highbp): mean birthwgt

How about across values of a categorical variable?

. codebook agegrp

. svy: mean birthwgt, over(agegrp)

Subpopulations

Getting fancy

We are now finished with this lesson.

. log close

Survey Data Analysis

Postestimation

Working with estimation results
Most standard postestimation commands support svy results:

estat

estimates

lincom, nlcom
predict, predictnl
test, testnl
margins, marginsplot, contrast, pwcompare

Survey Data Analysis

Postestimation

Survey specific features in estat

Archer-Lemeshow goodness-of-fit
estat gof

Coefficient of variation
estat cv

Design and misspecification effects
estat effects

estat lceffects

Survey design characteristics
estat svyset

Survey Data Analysis

Postestimation

Marginal effects
Predictive margins and marginal effects

margins

Graph results from margins

marginsplot

Peform ANOVA-style tests on the effects of factor variables
contrast

Perform pairwise comparisons of marginal means and slopes
pwcompare

Survey Data Analysis

Postestimation

Goodness of fit

Recall the logistic model we fit using the NHANES2 data.

. webuse nhanes2, clear

. local model highbp height weight c.age##c.age i.female i.race

. svy: logit `model´, baselevel

From the Archer-Lemeshow goodness-of-fit test we find no
evidence for lack of fit.

. estat gof

Logistic model for highbp, goodness-of-fit test

F(9,23) = 1.08
Prob > F = 0.4141

Survey Data Analysis

Postestimation

Predictive margins

Predictive margins
The variable race is a categorical (factor) variable with
three coded levels.

. describe race

. label list race

Let’s use margins to look at the predicted probablities of
high blood pressure for each level of race.

. margins race, vce(unconditional)

. marginsplot

Postestimation

Predictive margins

Survey Data Analysis

Postestimation

Marginal effects

Marginal effects
We can use the dydx() option to get margins to compute
the marginal effects of race.

. margins, vce(unconditional) dydx(race)

. marginsplot

Postestimation

Marginal effects

Survey Data Analysis

Postestimation

Discrete marginal effects

Contrasts
The contrast command and contrast operators are new
in Stata 12.
margins has a richer set of operators for computing
discrete marginal effects.

. margins r.race, vce(unconditional)

. marginsplot

Here is a profile plot corresponding to these marginal
effects.

Postestimation

Discrete marginal effects

Survey Data Analysis

Summary

1 Use svyset to specify the survey design for your data.
2 Use svydes to find strata with a single PSU.
3 Choose your variance estimation method; you can svyset

it.
4 Use the svy prefix with estimation commands.
5 Use subpop() instead of if and in.
6 Most standard postestimation commands support svy

results.

Summary

